G13ASF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G13ASF is a diagnostic checking routine suitable for use after fitting a Box–Jenkins ARMA model to a univariate time series using G13AEF or G13AFF. The residual autocorrelation function is returned along with an estimate of its asymptotic standard errors and correlations. Also, G13ASF calculates the Box–Ljung portmanteau statistic and its significance level for testing model adequacy.

2 Specification

```
 SUBROUTINE G13ASF(N, V, MR, M, PAR, NPAR, ISHOW, R, RCM, IRCM,
1                  CHI, IDF, SIGLEV, IW, LIW, WORK, LWORK, IFAIL)
 INTEGER           N, MR(7), M, NPAR, ISHOW, IRCM, IDF, IW(LIW),
1                  LIW, LWORK, IFAIL
 real              V(N), PAR(NPAR), R(M), RCM(IRCM,M), CHI, SIGLEV,
1                  WORK(LWORK)
```

3 Description

Consider the univariate multiplicative autoregressive-moving average model

$$\phi(B)\Phi(B^s)(W_t - \mu) = \theta(B)\Theta(B^s)\epsilon_t \tag{1}$$

where W_t, for $t = 1, 2, \ldots, n$ denotes a time series and ϵ_t, for $t = 1, 2, \ldots, n$ is a residual series assumed to be normally distributed with zero mean and variance σ^2 (> 0). The ϵ_t's are also assumed to be uncorrelated. Here μ is the overall mean term, s is the seasonal period and B is the backward shift operator such that $B^r W_t = W_{t-r}$. The polynomials in (1) are defined as follows:

$$\phi(B) = 1 - \phi_1 B - \phi_2 B^2 - \ldots - \phi_p B^p$$

is the non-seasonal autoregressive (AR) operator;

$$\theta(B) = 1 - \theta_1 B - \theta_2 B^2 - \ldots - \theta_q B^q$$

is the non-seasonal moving average (MA) operator;

$$\Phi(B^s) = 1 - \Phi_1 B^s - \Phi_2 B^{2s} - \ldots - \Phi_P B^{Ps}$$

is the seasonal AR operator; and

$$\Theta(B^s) = 1 - \Theta_1 B^s - \Theta_2 B^{2s} - \ldots - \Theta_Q B^{Qs}$$

is the seasonal MA operator. The model (1) is assumed to be stationary, that is the zeros of $\phi(B)$ and $\Phi(B^s)$ are assumed to lie outside the unit circle. The model (1) is also assumed to be invertible, that is the zeros of $\theta(B)$ and $\Theta(B^s)$ are assumed to lie outside the unit circle. When both $\Phi(B^s)$ and $\Theta(B^s)$ are absent from the model, that is when $P = Q = 0$, then the model is said to be non-seasonal.

The estimated residual autocorrelation coefficient at lag l, $\hat{r}_l$, is computed as:

$$\hat{r}_l = \frac{\sum_{t=l+1}^{n} (\hat{\epsilon}_{t-l} - \bar{\epsilon})(\hat{\epsilon}_t - \bar{\epsilon})}{\sum_{t=1}^{n} (\hat{\epsilon}_t - \bar{\epsilon})^2}, \; l = 1, 2, \ldots$$

where $\hat{\epsilon}_t$ denotes an estimate of the tth residual, ϵ_t, and $\bar{\epsilon} = \sum_{t=1}^{n} \hat{\epsilon}_t / n$. A portmanteau statistic, $Q_{(m)}$, is calculated from the formula (see Box and Ljung [1]):

$$Q_{(m)} = n(n+2) \sum_{l=1}^{m} \hat{r}_l^2 / (n - l)$$

where m denotes the number of residual autocorrelations computed. (Advice on the choice of m is given in Section 8.2.) Under the hypothesis of model adequacy, $Q_{(m)}$ has an asymptotic χ^2 distribution on $m-p-q-P-Q$ degrees of freedom. Let $\hat{r}^T = (\hat{r}_1, \hat{r}_2, \ldots, \hat{r}_m)$ then the variance-covariance matrix of $\hat{r}$ is given by:

$$\mathrm{Var}(\hat{r}) = [I_m - X(X^TX)^{-1}X^T]/n.$$

The construction of the matrix X is discussed in McLeod [2]. (Note that the mean, μ, and the residual variance, σ^2, play no part in calculating $\mathrm{Var}(\hat{r})$ and therefore are not required as input to G13ASF.)

Note. For additive models with fixed parameter values (i.e., fitted by G13DCF) G13DSF should be used instead of G13ASF.

4 References

[1] Box G E P and Ljung G M (1978) On a measure of lack of fit in time series models *Biometrika* **65** 297–303

[2] McLeod A I (1978) On the distribution of the residual autocorrelations in Box–Jenkins models *J. Roy. Statist. Soc. Ser. B* **40** 296–302

5 Parameters

1: N — INTEGER *Input*

On entry: the number of observations in the residual series, n.

If G13ASF is used following a call to G13AEF then N must be the value ICOUNT(2) returned by G13AEF.

If G13ASF is used following a call to G13AFF then N must be the value NRES returned by G13AFF.

Constraint: $\mathrm{N} \geq 3$.

2: V(N) — ***real*** array *Input*

On entry: V(t) must contain an estimate of ϵ_t, for $t = 1, 2, \ldots, n$.

If G13ASF is used following a call to G13AEF then the actual argument V must be EXR(ICOUNT(1) + 1) as returned by G13AEF.

If G13ASF is used following a call to G13AFF then the actual argument V must be RES as returned by G13AFF.

Constraint: V must contain at least two distinct elements.

3: MR(7) — INTEGER array *Input*

On entry: the orders vector (p, d, q, P, D, Q, s) as supplied to G13AEF or G13AFF.

Constraints:

$p, q, P, Q, s \geq 0$,
$p + q + P + Q > 0$,
if $s = 0$, then $P = 0$ and $Q = 0$.

4: M — INTEGER *Input*

On entry: the value of m, the number of residual autocorrelations to be computed. See Section 8.2 for advice on the value of M.

Constraint: $\mathrm{NPAR} < \mathrm{M} < \mathrm{N}$.

5: PAR(NPAR) — ***real*** array *Input*

On entry: the parameter estimates in the order $\phi_1, \phi_2, \ldots, \phi_p$, $\theta_1, \theta_2, \ldots, \theta_q$, $\Phi_1, \Phi_2, \ldots, \Phi_P$, $\Theta_1, \Theta_2, \ldots, \Theta_Q$.

Constraint: the elements in PAR must satisfy the stationarity and invertibility conditions.

6: NPAR — INTEGER *Input*

On entry: the total number of ϕ, θ, Φ and Θ parameters, i.e., $\mathrm{NPAR} = p + q + P + Q$.

Constraint: $\mathrm{NPAR} = \mathrm{MR}(1) + \mathrm{MR}(3) + \mathrm{MR}(4) + \mathrm{MR}(6)$.

7: ISHOW — INTEGER *Input*

On entry: ISHOW must be non-zero if the residual autocorrelations, their standard errors and the portmanteau statistics are to be printed and zero otherwise.

These quantities are available also as output variables in R, RCM, CHI, IDF and SIGLEV.

8: R(M) — ***real*** array *Output*

On exit: an estimate of the residual autocorrelation coefficient at lag l, for $l = 1, 2, \ldots, m$. If $\mathrm{IFAIL} = 3$ on exit then all elements of R are set to zero.

9: RCM(IRCM,M) — ***real*** array *Output*

On exit: the estimated standard errors and correlations of the elements in the array R. The correlation between $\mathrm{R}(i)$ and $\mathrm{R}(j)$ is returned as $\mathrm{RCM}(i,j)$ except that if $i = j$ then $\mathrm{RCM}(i,j)$ contains the standard error of $\mathrm{R}(i)$. If on exit, $\mathrm{IFAIL} \geq 5$, then all off-diagonal elements of RCM are set to zero and all diagonal elements are set to $1/\sqrt{n}$.

10: IRCM — INTEGER *Input*

On entry: the first dimension of the array RCM as declared in the (sub)program from which G13ASF is called.

Constraint: $\mathrm{IRCM} \geq \mathrm{M}$.

11: CHI — ***real*** *Output*

On exit: the value of the portmanteau statistic, $Q_{(m)}$. If $\mathrm{IFAIL} = 3$ on exit then CHI is returned as zero.

12: IDF — INTEGER *Output*

On exit: the number of degrees of freedom of CHI.

13: SIGLEV — ***real*** *Output*

On exit: the significance level of CHI based on IDF degrees of freedom. If $\mathrm{IFAIL} = 3$ on exit then SIGLEV is returned as one.

14: IW(LIW) — INTEGER array *Workspace*

15: LIW — INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which G13ASF is called.

Constraint: $\mathrm{LIW} \geq \max(\mathrm{MR}(1), \mathrm{MR}(3), \mathrm{MR}(4), \mathrm{MR}(6))$.

16: WORK(LWORK) — ***real*** array *Workspace*

17: LWORK — INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13ASF is called.

Constraint:

$$\begin{aligned} \mathrm{LWORK} \geq \;& \mathrm{NPAR} \times (\mathrm{M} + \mathrm{NPAR} + 1) + \max(\mathrm{MR}(1), \mathrm{MR}(3), \mathrm{MR}(4), \mathrm{MR}(6)) \times \\ & \max(\mathrm{MR}(7), 1) + \mathrm{M}. \end{aligned}$$

18: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL ≠ 0 on exit, users are recommended to set IFAIL to −1 before entry. **It is then essential to test the value of IFAIL on exit**.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, MR(1) < 0,
or MR(3) < 0,
or MR(4) < 0,
or MR(6) < 0,
or MR(7) < 0,
or MR(7) = 0 and either MR(4) > 0 or MR(6) > 0,
or MR(1) = MR(3) = MR(4) = MR(6) = 0,
or M ≤ NPAR,
or M ≥ N,
or NPAR ≠ MR(1) + MR(3) + MR(4) + MR(6),
or IRCM < M,
or LIW is too small,
or LWORK is too small.

IFAIL = 2

On entry, the autoregressive (or moving average) parameters are extremely close to or outside the stationarity (or invertibility) region. To proceed, the user must supply different parameter estimates in the array PAR.

IFAIL = 3

On entry, the residuals are practically identical giving zero (or near zero) variance. In this case CHI is set to zero, SIGLEV to one and all the elements of R set to zero.

IFAIL = 4

This is an unlikely exit brought about by an excessive number of iterations being needed to evaluate the zeros of the AR or MA polynomials. All output parameters are undefined.

IFAIL = 5

On entry, one or more of the AR operators has a factor in common with one or more of the MA operators. To proceed, this common factor must be deleted from the model. In this case, the off-diagonal elements of RCM are returned as zero and the diagonal elements set to $1/\sqrt{n}$. All other output quantities will be correct.

IFAIL = 6

This is an unlikely exit. At least one of the diagonal elements of RCM was found to be either negative or zero. In this case all off-diagonal elements of RCM are returned as zero and all diagonal elements of RCM set to $1/\sqrt{n}$.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

8.1 Timing

The time taken by the routine depends upon the number of residual autocorrelations to be computed, m.

8.2 Choice of m

The number of residual autocorrelations to be computed, m should be chosen to ensure that when the ARMA model (1) is written as either an infinite order autoregressive process:

$$\text{i.e.,}\quad W_t - \mu = \sum_{j=1}^{\infty} \pi_j (W_{t-j} - \mu) + \epsilon_t$$

or as an infinite order moving average process:

$$\text{i.e.,}\quad W_t - \mu = \sum_{j=1}^{\infty} \psi_j \epsilon_{t-j} + \epsilon_t$$

then the two sequences $\{\pi_1, \pi_2, \ldots\}$ and $\{\psi_1, \psi_2, \ldots\}$ are such that π_j and ψ_j are approximately zero for $j > m$. An over-estimate of m is therefore preferable to an under-estimate of m. In many instances the choice $m = 10$ will suffice. In practice, to be on the safe side, the user should try setting $m = 20$.

8.3 Approximate Standard Errors

When IFAIL is returned as 5 or 6 all the standard errors in RCM are set to $1/\sqrt{n}$. This is the asymptotic standard error of $\hat{r}_l$ when all the autoregressive and moving average parameters are assumed to be known rather than estimated.

8.4 Alternative Applications

G13ASF may be used for diagnostic checking of suitable univariate ARMA model, as described in Section 3, fitted by G13BEF or G13DCF. Great care must be taken in obtaining the input values for G13ASF from the output values from G13BEF or G13DCF.

9 Example

A program to fit an ARIMA(1,1,2) model to a series of 30 observations. 10 residual autocorrelations are computed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G13ASF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NPARMX, NPPCMX, NXMAX, IRES, LIWORK, LWORK, MMAX,
     +                 ICM, IRCM
      PARAMETER        (NPARMX=10,NPPCMX=NPARMX+1,NXMAX=50,IRES=550,
     +                 LIWORK=10,LWORK=500,MMAX=10,ICM=NPARMX+1,
```

```
     +                 IRCM=MMAX)
*     .. Local Scalars ..
      real             C, CHI, S, SIGLEV
      INTEGER          I, IDF, IFAIL, ISHOW, ITC, KFC, KPIV, M, NDF,
     +                 NIT, NPAR, NPPC, NRES, NST, NX
*     .. Local Arrays ..
      real             CM(ICM,ICM), PAR(NPARMX), R(MMAX),
     +                 RCM(IRCM,MMAX), RES(IRES), SD(NPPCMX), ST(NXMAX),
     +                 WORK(LWORK), X(NXMAX)
      INTEGER          ISF(4), IWORK(LIWORK), MR(7)
*     .. External Subroutines ..
      EXTERNAL         G13AFF, G13ASF, X04ABF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13ASF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX
      IF (NX.GT.0 .AND. NX.LE.NXMAX) THEN
         READ (NIN,*) (X(I),I=1,NX)
         READ (NIN,*) (MR(I),I=1,7)
         NPAR = MR(1) + MR(3) + MR(4) + MR(6)
         IF (NPAR.GT.0 .AND. NPAR.LE.NPARMX) THEN
            DO 20 I = 1, NPAR
               PAR(I) = 0.0e0
   20       CONTINUE
            KFC = 1
            C = 0.0e0
            NPPC = NPAR + KFC
*           * Set KPIV to 1 to obtain monitoring information *
            KPIV = 0
            NIT = 50
            IFAIL = 1
*
            CALL G13AFF(MR,PAR,NPAR,C,KFC,X,NX,S,NDF,SD,NPPC,CM,ICM,ST,
     +                  NST,KPIV,NIT,ITC,ISF,RES,IRES,NRES,IFAIL)
*
            IF (IFAIL.NE.0) WRITE (NOUT,99999) 'G13AFF fails. IFAIL =',
     +          IFAIL
*
            IF (IFAIL.EQ.0 .OR. IFAIL.EQ.9) THEN
               CALL X04ABF(1,NOUT)
               M = 10
               ISHOW = 1
               IFAIL = -1
*
               CALL G13ASF(NRES,RES,MR,M,PAR,NPAR,ISHOW,R,RCM,IRCM,CHI,
     +                     IDF,SIGLEV,IWORK,LIWORK,WORK,LWORK,IFAIL)
*
               IF (IFAIL.NE.0) WRITE (NOUT,99999)
     +             'G13ASF fails. IFAIL =', IFAIL
            END IF
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I2)
      END
```

9.2 Program Data

```
G13ASF Example Program Data
30                    : NX, length of the time series
 -217 -177 -166 -136 -110  -95  -64  -37
  -14  -25  -51  -62  -73  -88 -113 -120
  -83  -33  -19   21   17   44   44   78
   88  122  126  114   85   64   : End of time series
1  1  2  0  0  0  0  : MR, orders vector of the model
```

9.3 Program Results

```
 G13ASF Example Program Results

 RESIDUAL AUTOCORRELATION FUNCTION
 ---------------------------------

 LAG  K      1      2      3      4      5      6      7
 R(K)      0.020 -0.040 -0.019  0.068 -0.143 -0.046 -0.205
 ST.ERROR  0.007  0.125  0.128  0.150  0.168  0.168  0.178
 ---------------------------------------------------------
 LAG  K      8      9     10
 R(K)     -0.108 -0.001 -0.058
 ST.ERROR  0.179  0.181  0.183
 ---------------------------------------------------------

 BOX - LJUNG PORTMANTEAU STATISTIC =      3.465
               SIGNIFICANCE LEVEL =      0.839
 (BASED ON   7 DEGREES OF FREEDOM)

 VALUE OF IFAIL PARAMETER ON EXIT FROM G13ASF =   0
```